# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=perl-crypt-dh
pkgver=0.07
pkgrel=18
pkgdesc='Perl/CPAN Module Crypt::DH: Diffie-Hellman key exchange system.'
_cpan_name="Crypt-DH-$pkgver"
url='http://search.cpan.org/~mithaldu/Crypt-DH-0.07/lib/Crypt/DH.pm'
license=('GPL' 'PerlArtistic')
arch=('any')
makedepends=('make')
depends=('perl' 'perl-math-bigint-gmp' 'perl-module-install')
source=("http://search.cpan.org/CPAN/authors/id/M/MI/MITHALDU/$_cpan_name.tar.gz")
sha512sums=('60e58403e513535463290255989e9f2263107aeae6831d51d761e18527b1745baa45c16f655452fd618a0fb0e98413d5ed68f3926247ab2ad2da4032863b8ee7')

build() {
  cd $_cpan_name

  PERL_MM_USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}

package() {
  cd $_cpan_name

  make pure_install DESTDIR="$pkgdir"

  cd $pkgdir

  find . -name '.packlist' -delete
  find . -name '*.pod'     -delete

  # fix conflict with perl-math-bigint-gmp
  rm -rf "$pkgdir/usr/share/man/"
}

